```
> restart;
> # Backwards Differentiation Formula
> sigma0:=w->w^s;
```

$$\sigma 0 := w \to w^s$$

```
> s:=6;
```

$$s := 6$$

```
> rho0:=sigma0(1+x)*log(x+1);
```

$$\rho 0 := (1+x)^6 \ln(1+x)$$

```
> rho1:=series(rho0,x,s+1);
```

$$\rho 1 := x + \frac{11}{2} x^2 + \frac{37}{3} x^3 + \frac{57}{4} x^4 + \frac{87}{10} x^5 + \frac{49}{20} x^6 + O(x^7)$$

```
> rho2:=convert(rho1,polynom);
```

$$\rho 2 := x + \frac{11}{2} x^2 + \frac{37}{3} x^3 + \frac{57}{4} x^4 + \frac{87}{10} x^5 + \frac{49}{20} x^6$$

```
> rho3:=expand(unapply(rho2,x)(w-1));
```

$$\rho 3 := \frac{1}{6} - \frac{6}{5} w + \frac{15}{4} w^2 + \frac{15}{2} w^4 - \frac{20}{3} w^3 - 6 w^5 + \frac{49}{20} w^6$$

```
> beta:=1/coeff(rho3,w,s);
```

$$\beta := \frac{20}{49}$$

```
> sigma:=beta*sigma0(w);
```

$$\sigma := \frac{20}{49} w^6$$

```
> rho:=beta*rho3;
```

$$\rho := \frac{10}{147} - \frac{24}{49} w + \frac{75}{49} w^2 + \frac{150}{49} w^4 - \frac{400}{147} w^3 - \frac{120}{49} w^5 + w^6$$

```
> R:=[solve(rho=0)];
```

$$R := [1, \mathrm{RootOf}(147\_Z^5 - 213\_Z^4 + 237\_Z^3 - 163\_Z^2 + 62\_Z - 10, \mathit{index} = 1),$$
$$\mathrm{RootOf}(147\_Z^5 - 213\_Z^4 + 237\_Z^3 - 163\_Z^2 + 62\_Z - 10, \mathit{index} = 2),$$
$$\mathrm{RootOf}(147\_Z^5 - 213\_Z^4 + 237\_Z^3 - 163\_Z^2 + 62\_Z - 10, \mathit{index} = 3),$$
$$\mathrm{RootOf}(147\_Z^5 - 213\_Z^4 + 237\_Z^3 - 163\_Z^2 + 62\_Z - 10, \mathit{index} = 4),$$
$$\mathrm{RootOf}(147\_Z^5 - 213\_Z^4 + 237\_Z^3 - 163\_Z^2 + 62\_Z - 10, \mathit{index} = 5)]$$

```
> seq(abs(evalf(R[i])),i=1..s);
```

1., 0.4061232669, 0.4740348577, 0.8633802679, 0.8633802679, 0.4740348577